

NOETHERIAN OPERATORS IN MACAULAY2

JUSTIN CHEN, YAIRON CID-RUIZ, MARC HÄRKÖNEN, ROBERT KRONE, ANTON LEYKIN

ABSTRACT. A primary ideal in a polynomial ring can be described by the variety it defines and a finite set of Noetherian operators, which are differential operators with polynomial coefficients. We implement both symbolic and numerical algorithms to produce such a description in various scenarios as well as routines for studying affine schemes through the prism of Noetherian operators and Macaulay dual spaces.

1. INTRODUCTION

The idea of describing ideals in polynomial rings via systems of differential operators was brewing since the beginning of the twentieth century. In [14], Macaulay brought forth the notion of an *inverse system*, a system of differential conditions that describes a *modular system* (a system of polynomials, or a polynomial ideal, in the modern language).

It was apparent to the contemporaries of Macaulay that a finite number of differential conditions should suffice to describe a 0-dimensional affine or projective scheme. In [8], Gröbner derived explicit characterizations for ideals that are primary to a rational maximal ideal [9, p. 174-178]. Moreover, he suggested that the same program can be carried out for any primary ideal [7, §1].

Despite this early algebraic interest, a complete description of primary ideals in terms of differential operators was first obtained by analysts in the *Fundamental Principle of Ehrenpreis and Palamodov* [6, 16]. At the core of the Fundamental Principle, one has the following theorem by Palamodov.

Theorem 1.1 (Palamodov). *Let R be a polynomial ring $R = \mathbb{C}[x_1, \dots, x_n]$ over the complex numbers, $P \subseteq R$ a prime ideal, and $Q \subseteq R$ a P -primary ideal. There exist differential operators $A_1, \dots, A_m \in R\langle \partial_{x_1}, \dots, \partial_{x_n} \rangle$ such that $Q = \{f \in R \mid A_i \bullet f \in P \text{ for } 1 \leq i \leq m\}$.*

Following the terminology of Palamodov, the differential operators A_1, \dots, A_m are commonly called *Noetherian operators for the P -primary ideal Q* . Subsequent algebraic and computational approaches to characterize primary ideals with the use of differential operators have been given in [1], [15], [5], [3]; and, most recently, in [2] and [4].

The purpose of this note is to present the Macaulay2 package `NoetherianOperators`, which implements the algorithms for Noetherian operators introduced in [2] and [4] as well as the algorithms for (Macaulay) *dual spaces* addressed in [13, 12, 11]. While some of these algorithms rely on exact symbolic computation, the others employ numerical approximations using paradigms of *numerical algebraic geometry*.

2. COMPUTING NOETHERIAN OPERATORS FROM IDEALS

In this section we discuss four algorithms that can be used to compute a set of Noetherian operators for a primary ideal. The main method in the `NoetherianOperators` package is `noetherianOperators`, which contains implementations of symbolic algorithms. Specifying a value for the option `Strategy` allows the user to choose the algorithm that is used. The method `numericalNoetherianOperators` implements a numerical algorithm, which can deal with approximate input. Our list of strategies includes the following four algorithms:

- (1) **Punctual Hilbert scheme:** the main idea of this algorithm is to use the *punctual Hilbert scheme* to parametrize primary ideals.
- (2) **Symbolic algorithm via dual spaces:** this algorithm computes Noetherian operators as bases of *Macaulay dual spaces*.
- (3) **Numerical algorithm via interpolation:** this algorithm interpolates Noetherian operators from their specializations at several general points sampled on the underlying variety.

- (4) **Hybrid symbolic/numerical:** this approach optimizes the approach in (2) by using information obtained from applying the numerical algorithm (3) at one general point on the underlying variety.

We now discuss each algorithm and illustrate its use in the package. Throughout the article, let \mathbb{K} denote a field of characteristic zero, $R = \mathbb{K}[x_1, \dots, x_n]$ a polynomial ring over \mathbb{K} , and $P \subseteq R$ a prime ideal in R . We typically use Q to denote a P -primary ideal, and I to denote a general (not necessarily primary) ideal which has P as a minimal prime.

To represent Noetherian operators, this package introduces a new type called `DiffOp`, representing elements in $R(\partial_{x_1}, \dots, \partial_{x_n})$. A `DiffOp` is a hash table, where each key-value pair represents a term of the differential operator (with keys given by monomials in R , corresponding to a monomial in $\partial_{x_1}, \dots, \partial_{x_n}$, whose value is the associated polynomial coefficient). `DiffOps` can be added and scaled, and also act on polynomials in R . The output of `noetherianOperators` is a list of `DiffOps`, and the output of `numericalNoetherianOperators` is a list of `InterpolatedDiffOps`, a type similar to `DiffOp`, but whose coefficients are rational functions instead of polynomials.

2.1. Punctual Hilbert scheme. The backbone of this algorithm is [4, Theorem 2.1], which can be seen as a “representation theorem” that parametrizes primary ideals via three closely related objects (points in the punctual Hilbert scheme, differentially closed vector spaces, and submodules of the Weyl-Noether module).

For a prime ideal P of codimension c , let \mathbb{F} be the field of fractions of the integral domain R/P . Up to a linear change of coordinates, we may (and do) assume that $\{x_{c+1}, \dots, x_n\}$ is a maximal independent set of variables modulo P , to simplify notation.

We now recall the steps of [4, Algorithm 8.1]. The main idea of this algorithm is to reduce the study of arbitrary P -primary ideals in R to a zero-dimensional setting over the function field \mathbb{F} . This reduction is made by parametrizing P -primary ideals with the punctual Hilbert scheme $\text{Hilb}^m(\mathbb{F}[[y_1, \dots, y_c]])$. This is a quasiprojective scheme over the function field \mathbb{F} . Its classical points are ideals of colength m in the local ring $\mathbb{F}[[y_1, \dots, y_c]]$. For more details regarding punctual Hilbert schemes the reader is referred to [10]. We define the inclusion map

$$\gamma : R \hookrightarrow \mathbb{F}[y_1, \dots, y_c], \quad \begin{array}{ll} x_i & \mapsto y_i + \overline{x_i}, & \text{for } 1 \leq i \leq c, \\ x_j & \mapsto \overline{x_j}, & \text{for } c+1 \leq j \leq n, \end{array}$$

where $\overline{x_i}$ denotes the class of x_i in \mathbb{F} , for $1 \leq i \leq n$. With this, we can give the following explicit bijective correspondence

$$\begin{array}{ccc} \left\{ \begin{array}{l} P\text{-primary ideals of } R \\ \text{with multiplicity } m \text{ over } P \end{array} \right\} & \longleftrightarrow & \left\{ \text{points in } \text{Hilb}^m(\mathbb{F}[[y_1, \dots, y_c]]) \right\} \\ Q & \longrightarrow & J = \langle y_1, \dots, y_c \rangle^m + \gamma(Q)\mathbb{F}[y_1, \dots, y_c] \\ Q = \gamma^{-1}(J) & \longleftarrow & J \end{array}$$

The method `mapToPunctualHilbertScheme` can be used to compute the point in $\text{Hilb}^m(\mathbb{F}[[y_1, \dots, y_c]])$ that corresponds to a P -primary ideal Q . For notational purposes, the method `mapToPunctualHilbertScheme` uses variables $\{hx_1, \dots, hx_c\}$ (by adding an h in front of each of $\{x_1, \dots, x_c\}$) instead of $\{y_1, \dots, y_c\}$. The following example illustrates the use of the method `mapToPunctualHilbertScheme`.

```
i1 : needsPackage "NoetherianOperators";
i2 : S = QQ[x_1, x_2, x_3];
i3 : Q = ideal(x_1^2, x_2^2, x_1-x_2*x_3);
i4 : mapToPunctualHilbertScheme Q
      2
o4 = ideal (hx  - x hx , hx )
      1   3  2   2
      /   S   \
o4 : Ideal of frac|-----|[hx ..hx ]
      |(x , x )|  1   2
      \  2   1 /
```

After computing the point $I \subseteq \text{Hilb}^m(\mathbb{F}[[y_1, \dots, y_c]])$ corresponding to a P -primary ideal Q of multiplicity m over P , the inverse system J^\perp of J is computed. Lastly, an \mathbb{F} -basis of J^\perp is lifted to a set of Noetherian operators for the ideal Q .

This is the default strategy used to compute Noetherian operators, when the input is a primary ideal. It can also be explicitly called by specifying `Strategy => "PunctualHilbert"`, as shown below:

```
i5 : noetherianOperators(Q, Strategy => "PunctualHilbert")
o5 = {1, x dx_1 + dx_2}
      3
```

2.2. Symbolic algorithm via dual spaces. The next algorithm to compute a set of Noetherian operators is a direct approach which reduces the problem to linear algebra. For convenience, write \mathbf{t} for a maximal set of independent variables modulo P , and $\mathbf{x} := \{x_1, \dots, x_n\} \setminus \mathbf{t}$ as the dependent variables. Then in the localization $S := \mathbb{K}(\mathbf{t})[\mathbf{x}]$ of R , the extension of I to S is zero-dimensional. If now I is a zero-dimensional P -primary ideal, then a set of Noetherian operators for I is the same as a basis for the dual space of I at P (as will be discussed in Section 4). This in turn can be computed as the kernel of a *Macaulay matrix*, which is a matrix over R/P with columns indexed by differential monomials and rows indexed by elements of I , whose entries are the result of applying a differential monomial to an element of I . As the numbers of rows and columns increase, the kernel eventually stabilizes, at which point the result is returned.

This is the default strategy used to compute Noetherian operators, when the input is a pair of ideals, the second of which should be a minimal prime of the first. It can also be explicitly called by specifying `Strategy => "MacaulayMatrix"`, as shown below:

```
i6 : needsPackage "K3Carpets";
i7 : I = carpet(2, 2, Characteristic => 0);
o7 : Ideal of QQ[x ..x , y ..y ]
      0 2 0 2

i8 : R = ring I;
i9 : noetherianOperators(I, Strategy => "MacaulayMatrix")
o9 = {1, 2y dy_0 + y dy_1}
      1 2
```

Calling `noetherianOperators` with a minimal prime ideal P of I as the second argument will compute a set of Noetherian operators for the P -primary component of I . If I is unmixed, then the result of applying this method to all associated primes can be thought of as dual to a primary decomposition of I .

```
i10 : (P1, P2) = (radical I, ideal(R_0 + R_1));
i11 : J = intersect(I, P2^2);
i12 : noetherianOperators(J, P1)
o12 = {1, 2y dy_0 + y dy_1}
      1 2

i13 : noetherianOperators(J, P2)
o13 = {1, dx_0}
```

2.3. Numerical algorithm via interpolation. We also provide algorithms to compute Noetherian operators purely from numerical data, bypassing the need to compute Gröbner bases. This is based on computing a set of *specialized Noetherian operators*, i.e. the result of evaluating (at some point) all polynomial coefficients in a set of Noetherian operators. The key observation is that one can obtain a set of specialized Noetherian operators by suitably slicing the variety [2, Theorem 4.1]. More precisely, for a P -primary ideal $Q \subseteq \mathbb{C}[\mathbf{t}, \mathbf{x}]$ and a point $p = (\mathbf{t}_0, \mathbf{x}_0) \in V(P)$, a minimal set of specialized Noetherian operators corresponds to a basis of the dual space of the zero-dimensional ideal $Q + (\mathbf{t} - \mathbf{t}_0)$ at the point p . The function `specializedNoetherianOperators` can be used to perform this computation.

```
i14 : p = point{{1,1,1,1,1,1}};
i15 : specializedNoetherianOperators(I, p, DependentSet => {R_1, R_3, R_4})
o15 = {1, 2*dy_0 + dy_1}
```

Once a set of specialized Noetherian operators has been computed at a single general point, subsequent computations at other points can be sped up as the monomial support of a valid set of Noetherian operators is known (this fact also underlies the hybrid approach in Section 2.4). After specialized Noetherian operators are computed at sufficiently many general points on the variety, the original set of Noetherian operators can be recovered from their specializations via interpolation of rational functions, cf. [2, Algorithm 5].

This is the preferred strategy when the input is inexact, although it can also be used for exact input, as shown below. Note that the value of `DependentSet` must be specified:

```

i16 : numericalNoetherianOperators(I, DependentSet => {R_1, R_3, R_4})
      x
      1
o16 = {1, 1*dy_0 + ---*dy_1}
      2x
      0

```

By default, `Bertini` is used to sample points on $V(\sqrt{I})$. The user can specify their own sampling function with the option `Sampler`. The sampler should be a function that takes an integer n and the ideal I as input, and return a List of n points on the variety.

```

i17 : needsPackage"NumericalImplicitization";
i18 : numericalNoetherianOperators(I, DependentSet => {R_1, R_3, R_4},
Sampler => (n,I) -> apply(n, i -> point sub(matrix realPoint radical I, CC)))
      x
      1
o18 = {1, 1*dy_0 + ---*dy_1}
      2x
      0

```

2.4. Hybrid symbolic/numerical. In Section 2.2, Noetherian operators are found by computing the kernel of a Macaulay matrix with entries in the function field $\mathbb{K}(\mathbf{t})$, but computations in this field can be expensive. The numerical approach in Section 2.3 instead specializes the independent variables to random values. This allows computations to be done in \mathbb{K} (typically with $\mathbb{K} = \mathbb{C}$) which is cheaper but the result consists of specializations of the Noetherian operators. A hybrid approach can combine the best of both strategies. In essence, the information revealed from running the numerical algorithm at a single point (without performing interpolation) can be used to trim the Macaulay matrix down to a smaller (optimal) size, without changing the kernel. For more details, we refer the interested reader to [2, Section 4.1].

This strategy is called by specifying `Strategy => "Hybrid"` with the method `noetherianOperators`. On larger examples, this strategy can greatly outperform the approach in Section 2.2.

```

i19 : noetherianOperators(I, Strategy=>"Hybrid")
o19 = {1, 2x y dy_0 + x y dy_1}
      0 2      2 1

```

As in the numerical algorithm, the user may also specify a sampling function to find a general point.

3. COMPUTING IDEALS FROM NOETHERIAN OPERATORS

In this section, we discuss a procedure that can be seen as the inverse of the process of computing a set of Noetherian operators. First, note that for any R -bisubmodule $\mathcal{E} \subseteq R\langle \partial_{x_1}, \dots, \partial_{x_n} \rangle$ of the Weyl algebra, the set

$$\{f \in R \mid A \bullet f \in P \text{ for all } A \in \mathcal{E}\}$$

is always a P -primary ideal in R (cf. [3, Proposition 3.5]). We now consider the following problem:

Given an R -bisubmodule $\mathcal{E} \subseteq R\langle \partial_{x_1}, \dots, \partial_{x_n} \rangle$, compute (generators for) the P -primary ideal $\{f \in R \mid A \bullet f \in P \text{ for all } A \in \mathcal{E}\}$.

This is accomplished by [4, Algorithm 8.2]. The idea is to use the explicit maps provided in [4, Theorem 2.1] in inverse order to how they appear in [4, Algorithm 8.1] (i.e., as discussed in Section 2.1). It should be noted that our implementation solves the following effective version of the problem above:

Given $A_1, \dots, A_m \in R\langle \partial_{x_1}, \dots, \partial_{x_n} \rangle$, compute the P -primary ideal $\{f \in R \mid A \bullet f \in P \text{ for all } A \in \mathcal{E}\}$, where $\mathcal{E} \subseteq R\langle \partial_{x_1}, \dots, \partial_{x_n} \rangle$ is the R -bisubmodule generated by A_1, \dots, A_m .

The function `getIdealFromNoetherianOperators` implements [4, Algorithm 8.2]. Below we show an example in which given a P -primary ideal Q , we compute a set of Noetherian operators for Q , and then we recover Q from its Noetherian operators along with P . In general, this process may result in a different set of generators for Q .

```

i20 : R = QQ[x_1,x_2,x_3];
i21 : Q = ideal(x_1^2, x_2^2, x_3^2, x_1*x_2 + x_1*x_3 +x_2*x_3);

```

```

i22 : L = noetherianOperators Q
o22 = {1, dx_3, dx_2, dx_1, - dx_1*dx_2 + dx_2*dx_3, - dx_1*dx_2 + dx_1*dx_3}
i23 : Q' = getIdealFromNoetherianOperators(L, radical Q)
      2      2      2
o23 = ideal (x , x , x x + x x + x x , x )
      3      2      1 2      1 3      2 3      1
i24 : Q == Q'
o24 = true

```

4. DUAL SPACES AND LOCAL HILBERT FUNCTIONS

In Section 2, dual spaces were used in service of computing Noetherian operators. However, dual spaces can also directly provide information about polynomial ideals. Suppose P is the maximal ideal corresponding to a \mathbb{K} -rational point $p \in \mathbb{K}^n$, and $I \subseteq R$ an ideal with $p \in V(I)$. The *dual space of I at P* is

$$D_P[I] := \{A \in \mathbb{K}[\partial_{x_1}, \dots, \partial_{x_n}] \mid (A \bullet f)(p) = 0 \text{ for all } f \in I\}.$$

The dual space is a subspace of the space of differential operators on R with constant coefficients and finite support which uniquely determines IR_P , where R_P denotes the localization of R at P .

The following dual space algorithms work with numerical data, for example if $\mathbb{K} = \mathbb{C}$ and the point associated to P is known only approximately. This is in contrast to symbolic algorithms relying on Gröbner bases. The methods described in this section were previously part of a package titled `NumericalHilbert` which has now been incorporated into `NoetherianOperators` due to an overlap in functionality.

If P is a minimal prime of I , then the dual space is finite dimensional, and a basis of the dual space is a set of Noetherian operators for the P -primary component of I . Otherwise the dual space is infinite dimensional, and only a truncation up to a specified degree can be computed. The methods `zeroDimensionalDual` and `truncatedDual` compute a basis for the dual space in these two cases. As in Section 2.2, these dual spaces are computed as kernels of Macaulay matrices. From a basis of the dual space truncated to degree d , it is straightforward to compute the local Hilbert function of R/I up to degree d , and this is implemented as applying `hilbertFunction` to a `DualSpace` object.

```

i25 : R = CC[x_1,x_2];
i26 : I = ideal{x_1^2 + x_2^2 - 4, (x_1 - 1)^2};
i27 : p = point{{1.0, 1.7320508}};
i28 : D = zeroDimensionalDual(p, I)
o28 = | 1 -1.73205x_1+x_2 |
o28 : DualSpace
i29 : apply(3, i -> hilbertFunction(i, D))
o29 = {1, 1, 0}
o29 : List

```

Another way of truncating the dual space of a positive dimensional ideal is with respect to a local elimination order instead of a degree order. In [12] this is referred to as an *eliminating dual space*. Assume $\{x_{c+1}, \dots, x_n\}$ is a maximal set of independent variables for R/I , and choose an order that eliminates $V = \{x_1, \dots, x_c\}$. An eliminating dual for I with respect to V truncated to degree d , denoted $E_P^d[I, V]$, is defined as the set of dual operators whose lead terms with respect to the monomial order have degree at most d in the variables V , and is computed with method `eliminatingDual`. When $V = \{x_1\}$, such a truncated dual space allows one to find the dual space of the colon ideal $I : x_1$ directly (without requiring the potentially expensive symbolic computation of finding $I : x_1$):

$$E_P^d[I : x_1, \{x_1\}] = x_1 \cdot E_0^{d+1}[I, \{x_1\}]$$

where $x_1 \cdot A$ represents the right action of $x_1 \in R$ on $A \in \mathbb{K}[\partial_{x_1}, \dots, \partial_{x_n}]$ (for example $x_1 \cdot \partial_{x_1}^2 = 2\partial_{x_1}$). A representation of these colon ideals are needed in [12, Algorithm 5.1] for identifying embedded primes on a curve.

Using [11, Algorithm 23], computing truncated dual spaces up to a certain degree provides a numerical algorithm for finding a full set of generators for the initial ideal of I with respect to a local degree order, and this is implemented by `gCorners`. In the process, a standard basis can be computed by specifying `ProduceSB => true`.

```

i30 : gCorners(point p, I)
o30 = | x_2 x_1^2 |
      1      2
o30 : Matrix R <--- R

```

Finding the initial ideal via approximate numerical methods is an essential part of `isPointEmbedded`, which implements a numerical algorithm for the detection of an embedded component developed in [13, Algorithm 4.2].

REFERENCES

- [1] G. Brumfiel. Differential operators and primary ideals. *J. Algebra*, 51(2):375–398, 1978.
- [2] Justin Chen, Marc Härkönen, Robert Krone, and Anton Leykin. Noetherian operators and primary decomposition. *arXiv preprint arXiv:2006.13881*, 2020.
- [3] Y. Cid-Ruiz. Noetherian operators, primary submodules and symbolic powers. *to appear in Collect. Math.*, 2019. arXiv:1909.07253.
- [4] Yairon Cid-Ruiz, Roser Homs, and Bernd Sturmfels. Primary ideals and their differential equations. *to appear in Foundations of Computational Mathematics*, 2020.
- [5] Alberto Damiano, Irene Sabadini, and Daniele C. Struppa. Computational methods for the construction of a class of Noetherian operators. *Experiment. Math.*, 16(1):41–53, 2007.
- [6] Leon Ehrenpreis. *Fourier analysis in several complex variables*. Pure and Applied Mathematics, Vol. XVII. Wiley-Interscience Publishers A Division of John Wiley & Sons, New York-London-Sydney, 1970.
- [7] W. Gröbner. La théorie des idéaux et la géométrie algébrique. In *Deuxième Colloque de Géométrie Algébrique, Liège, 1952*, pages 129–144. Georges Thone, Liège; Masson & Cie, Paris, 1952.
- [8] Wolfgang Gröbner. Über eine neue idealtheoretische Grundlegung der algebraischen Geometrie. *Math. Ann.*, 115(1):333–358, 1938.
- [9] Wolfgang Gröbner. *Algebraische Geometrie. 2. Teil: Arithmetische Theorie der Polynomringe*. Bibliographisches Institut, Mannheim-Vienna-Zurich, 1970.
- [10] A. Iarrobino. Punctual Hilbert schemes. *Bull. Amer. Math. Soc.*, 78:819–823, 1972.
- [11] Robert Krone. Numerical algorithms for dual bases of positive-dimensional ideals. *Journal of Algebra and Its Applications*, 12(06):1350018, 2013.
- [12] Robert Krone and Anton Leykin. Eliminating dual spaces. *Journal of Symbolic Computation*, 79:609–622, 2017.
- [13] Robert Krone and Anton Leykin. Numerical algorithms for detecting embedded components. *Journal of Symbolic Computation*, 82:1–18, 2017.
- [14] F. S. Macaulay. *The algebraic theory of modular systems*. Cambridge Mathematical Library. Cambridge University Press, Cambridge, 1994. Revised reprint of the 1916 original, With an introduction by Paul Roberts.
- [15] Ulrich Oberst. The construction of Noetherian operators. *J. Algebra*, 222(2):595–620, 1999.
- [16] V. P. Palamodov. *Linear differential operators with constant coefficients*. Translated from the Russian by A. A. Brown. Die Grundlehren der mathematischen Wissenschaften, Band 168. Springer-Verlag, New York-Berlin, 1970.

SCHOOL OF MATHEMATICS, GEORGIA INSTITUTE OF TECHNOLOGY, ATLANTA, GEORGIA

Email address: {justin.chen,leykin}@math.gatech.edu

Email address: harkonen@gatech.edu

DEPARTMENT OF MATHEMATICS: ALGEBRA AND GEOMETRY, GHENT UNIVERSITY, KRIJGSLAAN 281 – S25, 9000 GENT, BELGIUM

Email address: Yairon.CidRuiz@UGent.be

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF CALIFORNIA, DAVIS, CALIFORNIA

Email address: rkrone@math.ucdavis.edu